
News  

Spike2 version 7.04 released 
The latest revision of Spike2 is now available and includes the following new features: 

• Spike2 can now sample data from Talker software modules that interface 
between arbitrary data capture devices and Spike2. 

• When sampling WaveMark data with a Power1401, Power1401 mk II or 
Micro1401-3 you can now reject spikes that are outside an amplitude range. 

• Reworked clustering dialog with more toolbar buttons and simpler keyboard 
shortcuts 

• Keyboard shortcuts for sampling control 

For a full list of new features, see the revision history in the on-line help. 

   

Future meetings and events 
9th International Conference on Neuroethology 
Palacio de Congresos, 
Salamanca, Spain 
August 2nd – 7th  2010 

Symposium for Chinese Neuroscientists Worldwide 2010 (SCNW 2010) 
Yu Quan Island Hotel, 
Nanchang, Jiangxi, China 
August 4th – 10th  2010 

Society for Neuroscience 2010 
San Diego Convention Center, 
San Diego, CA, USA 
November 13th – 17th  2010 

US Training Days at SFN 
Our US training days will take place on the 11th and 12th of November, prior to the Society 
for Neuroscience meeting in San Diego. The venue and booking details will be 
announced in due course. If you would like to be added to our mailing list for training day 
announcements please contact sales@ced.co.uk. 

Latest versions of Spike2 and Signal 
Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Released Signal downloads Released 
Spike2 version 7.04         07/10 Signal version 4.07         01/10 

Spike2 version 6.14        03/10 Signal version 3.13         05/10 

Spike2 demo Signal demo 

Spike2   

 

Q. I would like to amalgamate data recorded from a separate acquisition device with 
data sampled using Spike2. Is there any way of achieving this?  

A. As of Spike2 version 7.04 it is possible to sample additional channels from 
custom software modules (Talkers) that interface between arbitrary data capture 
devices and Spike2. These additional channels can be sampled to the same file 
as data being recorded using a 1401 interface. As an example, consider a motion 
tracking system that outputs co-ordinate values to a PC via a suitable connection. 
This output could be added automatically to a sampling Spike2 file by writing a 
Talker interface. 

 Three example Talkers are included with Spike2. To install these, along with 
documentation describing how to write your own custom modules, select the 
Custom setup option during installation of Spike2 version 7.04 and check the 
Talker documentation option. Channels from Talker modules are added to the 
sampling configuration from the drop-down menu next to the New button. 

 
 

  

 

 

Sampling configuration with TalkerEx interface enabled 

 We make sufficient information available to enable manufacturers of other data 
capture devices will produce their own Talker modules, but do feel free to contact 
us if you have a particular device that you would like to interface with Spike2. 

Spike2 is also able to accommodate a wide range of data recorded with other 
systems using the File menu Import function. It is also possible to amalgamate 
imported data into existing data files using the MergeFiles.s2s script, which is 
included in the Scripts folder of your Spike2 installation directory. 

 Further information on Talker interfaces can be found in the version 7.04 on-line 
help and in the Talker folder of the main Spike2 folder in the Start menu 
Programs section (if requested during installation).  

 

 

  

Q. I am using Signal to record evoked response data and need to generate an 
output pulse before the sampling sweep to arm a stimulator prior to triggering. 
Fixed interval mode can do this, but sets the sweeps to be always internally 
timed at the selected interval. I would like to be able to trigger the sweeps 

CED
You can print the sampling configuration information to the Log window or Copy As Text to the clipboard



manually during the recording based on my assessment of the subject. Is there a 
mode that allows this? 

A. In Outputs frame mode it is possible to set-up the sampling sweep to start at a 
specified point during the outputs frame and set pulses before, after and during 
the captured data sweep. The main difference between this mode and Fixed 
interval is that the outputs sweep can be externally triggered. 

 The pulse outputs and sampling sweep time are defined in the Pulses 
configuration dialog using the Frame and Trigger fields. The Trigger field 
sets where the sampling sweep occurs in the outputs frame (highlighted below). 
The external trigger pulse will trigger the full length of the output sweep, including 
any pulses outside the time range of the sampling sweep. 

 
Pulse configuration dialog in Outputs frame mode 

Scripts: Spike2  

Q. Is there a way to export the instantaneous frequency values of a WaveMark 
channel? 

A. The attached script, DumpInstFreq.s2s, calculates the instantaneous 
frequency values from a nominated event, marker or WaveMark channel and 
writes them to the Log window. You can also display these data types as an 
instantaneous frequency plot using the Channel Draw Mode functions, 
available from the View menu in Spike2. 

Scripts: Signal  
Q. I have two files, one containing a control experiment and one containing a 

response to a drug preparation, and would like to subtract one from the other. 

A. The attached script, SubFileData.sgs, can be used to subtract data in one file 
from another on a frame by frame basis and show the result in a new memory 
view. The data files should include the same number of channels and be 
sampled at the same rate. 

Did you know…? 
You can add scripts to the Script Bar in Spike2 and Signal so that they can be run with a 
single button click. Select Script Bar List.. from the Script menu to add a script 
and then select the Script Bar option from the same menu to display the toolbar. 

Recent questions   
Q. Would it be possible to control the writing to disk of data in Spike2 remotely? 

Ideally we’d like to do this with a keyboard situated away from the sampling PC.  

A. The following keyboard shortcuts for sampling commands are now available in 
Spike2 version 7.04, allowing the use of a remote keyboard to control sampling. 

• Sample Start/Stop – Ctrl+Alt+S 
• Sample Reset – Ctrl+Alt+R 
• Sample Abort – Ctrl+Alt+A 
• Toggle Write to disk – Shift+Ctrl+W 

'DumpInstFreq|Script to write instantaneous frequency values as text to the Log window

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 19/07/10

var evChan%;
var numEvents%;
var data%;
var sTime,eTime;
var a%;
var num%;

if ViewKind() = 0 then              'Check we have a data file open
data%:= View();
FrontView(data%);
else
data%:= FileOpen("",0);        'If not prompt user to select one
WindowVisible(1);
endif;

if data% < 0 then
    Message("Could not open a data file!");     'Quit if no file opened
    halt;endif;


View(LogHandle());
Window(0,0,50,100);
WIndowVisible(0);
EditSelectAll();
EditClear();
View(data%);

DlgCreate("Spike channel selection");
DlgChan(1,"Select spike channel",30);
DlgShow(evChan%);

CursorSet(2);
Interact("Position cursors around area to extract intervals",1023);
CursorRenumber();
sTime:= Cursor(1);
eTime := Cursor(2);
num% := Count(evChan%,sTime,eTime);
if num% > 0 then
    DumpFrequencies(num%+1);
endif;


Proc DumpFrequencies(numEv%)
var evTimes[numEv%];
var inst[numEv%];
numEvents% := ChanData(evChan%,evTimes[1:],sTime,eTime);                            'Put event times into array
if Count(evChan%,0,sTime) > 0 then                                                  'Fill first array element with time of event before start period
    evTimes[0] := LastTime(evChan%,sTime);                                          'If first event, put zero in first element
else
    evTimes[0] := 0;
endif;
ArrConst(inst[],evTimes[]);
ArrDiff(inst[]);                                                      'Convert time array to intervals
ArrDivR(inst[],1);                                                    'Convert to instantaneous frequencies
PrintLog("Event times and instantaneous frequencies between %g s and %g s on channel %d\n\n",sTime,eTime,evChan%);
PrintLog("Time\tInst freq\n");
for a% := 1 to numEv% -1 do
	PrintLog("%g\t%g\n",evTimes[a%],inst[a%]);
next;
FrontView(LogHandle());
end;




CED
DumpInstFreq.s2s


'$SubFileData|Subtract one data file from another on a frame by frame basis. The data files must contain the 
'same no. of channels and be sampled at the same rate.

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application. 

'CED 20/07/10

var file1%;		'handle for master file
var file2%;		'handle for file to subtract
var noFrames1%;	'frame count from master file
var noFrames2%;	'frame count from file to subtract	
var stepFrames%;
var c%;			'counter for frames
var ch%;           'counter for channels
var dupFile%;      'Handle of new file
var noChans1%;
var noChans2%;
var noPoints1%,noPoints2%;
var gFloat%[20];              'global for floating window states

HideAll();
file1%:= FileOpen("",0,0,"Select control file");		'Prompt user to open master file
noFrames1%:= FrameCount();		'Get frame count
noChans1%:=ChanCount(-4);	'Count channels
noPoints1%:=ChanPoints(1);

file2%:= FileOpen("",0,0,"Select file to subtract");	'Prompt user to open file to subtract
noFrames2%:= FrameCount();		'Get frame count
noChans2%:= ChanCount(-4);
noPoints2%:=ChanPoints(1);

if noChans1% <> noChans2% or noPoints1% <> noPoints2% then
    Message("Selected files should have the same no. of channels and data points\nRun script again and select different files");
    RestoreAll();
    Halt;endif;

if noFrames1% > noFrames2% then
    stepFrames%:= noFrames2%
else
    stepFrames%:= noFrames1%
endif;
View(file1%);
dupFile%:=SetCopy(-1,"Subtracted data",0);					'Make duplicate of view

for c%:=1 to stepFrames% do
    View(dupFile%).Frame(c%);
    View(file1%).Frame(c%);
   for ch%:=1 to noChans1% do
        ArrConst(View(dupFile%,ch%).[], View(file1%,ch%).[]);
        next;
    
    if c% < stepFrames% then
        View(dupFile%).AppendFrame();
    endif;
next;    

for c%:=1 to stepFrames% do		'For each frame from file 2
    View(dupFile%).Frame(c%);
    View(file2%).Frame(c%);
    for ch%:=1 to noChans1% do
        ArrSub(View(dupFile%,ch%).[], View(file2%,ch%).[]);
        next;
        
next;

View(dupFile%).WindowVisible(1);
RestoreAll();

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end


CED
SubFileData.sgs

CED
Multiple states enable you to vary the sampling sweep start time and pulse settings between states



These key combinations have been made deliberately awkward to prevent 
accidental activation. 

User group      
Try the CED Forums bulletin board for software and hardware support 

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

